NAG Library Routine Document

C02AHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C02AHF determines the roots of a quadratic equation with complex coefficients.

2 Specification

```
SUBROUTINE C02AHF (AR, AI, BR, BI, CR, CI, ZSM, ZLG, IFAIL)

INTEGER             IFAIL
REAL (KIND=nag_wp)  AR, AI, BR, BI, CR, CI, ZSM(2), ZLG(2)
```

3 Description

C02AHF attempts to find the roots of the quadratic equation $az^2 + bz + c = 0$ (where a, b and c are complex coefficients), by carefully evaluating the 'standard' closed formula

$$z = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}.$$

It is based on the routine CQDRTC from Smith (1967).

Note: it is not necessary to scale the coefficients prior to calling the routine.

4 References

Smith B T (1967) ZERPOL: a zero finding algorithm for polynomials using Laguerre's method *Technical Report* Department of Computer Science, University of Toronto, Canada

5 Parameters

1: AR – REAL (KIND=nag_wp) *Input*
2: AI – REAL (KIND=nag_wp) *Input*

On entry: AR and AI must contain the real and imaginary parts respectively of a, the coefficient of z^2.

3: BR – REAL (KIND=nag_wp) *Input*
4: BI – REAL (KIND=nag_wp) *Input*

On entry: BR and BI must contain the real and imaginary parts respectively of b, the coefficient of z.

5: CR – REAL (KIND=nag_wp) *Input*
6: CI – REAL (KIND=nag_wp) *Input*

On entry: CR and CI must contain the real and imaginary parts respectively of c, the constant coefficient.

7: ZSM(2) – REAL (KIND=nag_wp) array *Output*

On exit: the real and imaginary parts of the smallest root in magnitude are stored in ZSM(1) and ZSM(2) respectively.

8: ZLG(2) – REAL (KIND=nag_wp) array *Output*

On exit: the real and imaginary parts of the largest root in magnitude are stored in $\text{ZLG}(1)$ and $\text{ZLG}(2)$ respectively.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $(\text{AR}, \text{AI}) = (0, 0)$. In this case, $\text{ZSM}(1)$ and $\text{ZSM}(2)$ contain the real and imaginary parts respectively of the root $-c/b$.

$\text{IFAIL} = 2$

On entry, $(\text{AR}, \text{AI}) = (0, 0)$ and $(\text{BR}, \text{BI}) = (0, 0)$. In this case, $\text{ZSM}(1)$ contains the largest machine representable number (see X02ALF) and $\text{ZSM}(2)$ contains zero.

$\text{IFAIL} = 3$

On entry, $(\text{AR}, \text{AI}) = (0, 0)$ and the root $-c/b$ overflows. In this case, $\text{ZSM}(1)$ contains the largest machine representable number (see X02ALF) and $\text{ZSM}(2)$ contains zero.

$\text{IFAIL} = 4$

On entry, $(\text{CR}, \text{CI}) = (0, 0)$ and the root $-b/a$ overflows. In this case, both $\text{ZSM}(1)$ and $\text{ZSM}(2)$ contain zero.

$\text{IFAIL} = 5$

On entry, $\tilde{b}$ is so large that $\tilde{b}^2$ is indistinguishable from $\tilde{b}^2 - 4\tilde{a}\tilde{c}$ and the root $-b/a$ overflows, where $\tilde{b} = \max(|\text{BR}|, |\text{BI}|)$, $\tilde{a} = \max(|\text{AR}|, |\text{AI}|)$ and $\tilde{c} = \max(|\text{CR}|, |\text{CI}|)$. In this case, $\text{ZSM}(1)$ and $\text{ZSM}(2)$ contain the real and imaginary parts respectively of the root $-c/b$.

If $\text{IFAIL} > 0$ on exit, then $\text{ZLG}(1)$ contains the largest machine representable number (see X02ALF) and $\text{ZLG}(2)$ contains zero.

7 Accuracy

If $\text{IFAIL} = 0$ on exit, then the computed roots should be accurate to within a small multiple of the ***machine precision*** except when underflow (or overflow) occurs, in which case the true roots are within a small multiple of the underflow (or overflow) threshold of the machine.

8 Further Comments

None.

9 Example

This example finds the roots of the quadratic equation $z^2 - (3.0 - 1.0i)z + (8.0 + 1.0i) = 0$.

9.1 Program Text

```
    Program c02ahfe

!     C02AHF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: c02ahf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)              :: ai, ar, bi, br, ci, cr
      Integer                         :: ifail
!     .. Local Arrays ..
      Real (Kind=nag_wp)              :: zlg(2), zsm(2)
!     .. Executable Statements ..
      Write (nout,*) 'C02AHF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) ar, ai, br, bi, cr, ci

      ifail = -1
      Call c02ahf(ar,ai,br,bi,cr,ci,zsm,zlg,ifail)

      If (ifail==0) Then
        Write (nout,*)
        Write (nout,*) 'Roots of quadratic equation'
        Write (nout,*)
        Write (nout,99999) 'z = ', zsm(1), zsm(2), '*i'
        Write (nout,99999) 'z = ', zlg(1), zlg(2), '*i'
      End If

99999 Format (1X,A,1P,E12.4,Sp,E14.4,A)
    End Program c02ahfe
```

9.2 Program Data

```
C02AHF Example Program Data
 1.0    0.0   -3.0    1.0    8.0    1.0       :AR  AI  BR  BI  CR  CI
```

9.3 Program Results

```
 C02AHF Example Program Results

 Roots of quadratic equation

 z =   1.0000E+00   +2.0000E+00*i
 z =   2.0000E+00   -3.0000E+00*i
```